
The MASTERMIND Game    Functions  

The MAIN ROW (B3:E3)

Instructions:  Create this pull down menu in B2, “smart copy” it to the other three cells, and then 
change the “=B2” in the list to match the respective cell (i.e. change them to =C2, =D2, and =E2).

Use DATA VALIDATION to make a list with the following 7 items:
1,2,3,4,5,6,=B2

Then in the INPUT MESSAGE, give a TITLE of,
Instructions:

and an INPUT MESSAGE of,
Enter a number between 1 and 6.

For All the Other Rows (B4:E14)

Use DATA VALIDATION to make a list with the following 6 items:
1,2,3,4,5,6

Then in the INPUT MESSAGE, give a TITLE of,
Instructions:

and an INPUT MESSAGE of,
Enter a number between 1 and 6.

The RANDOM ROW (B2:E2)

Instructions:  Enter the first line below into B2 and “smart copy” it to the other three cells.

=IF($I$2="OFF",B$2,RANDBETWEEN(1,6))

The Random BUTTON (I2)

Use DATA VALIDATION to make a list with the following two items:
Random ON,OFF

Then in the INPUT MESSAGE, give a TITLE of,
Instructions:

and an INPUT MESSAGE of,
Set to "Random ON" to generate a random set of main numbers. Then set to "OFF" and manually 
enter =B2, =C2, =D2, =E2 into the four blue cells.

Blacks Functions (K4:N14)

Instructions: Copy the function below into K4 and then “smart copy” to the next columns of blacks 
(to N4). Then copy the first blacks row to the rest of the blacks rows (down to the bottom of your game-
board.)

=IF(B4=B$3,1,0)

Pre-whites Functions (O4:R14)

Instructions: Copy the function below into O4 and then “smart copy” to the next columns of pre-whites 
(to R4) and change three $X's in the respective “X” copy to correct the 3 column names. Then copy the 
first pre-white row to the rest of the pre-white rows (down to the bottom of your game-board.)

=IF(B$3=B4,0,IF(AND(B4=$C$3, NOT($C$3=$C4)),1,IF(AND(B4=$D$3, NOT($D$3=$D4)),1, 
IF(AND(B4=$E$3, NOT($E$3=$E4)),1,0))))

Multiples Functions (S4:V14)

Instructions: Copy the function below into S4 and then “smart copy” to the next columns of 
multiples (to V4). Then copy the first multiples row to the rest of the multiples rows (down to the 
bottom of your game-board.)

=MIN(COUNTIF($B$3:$E$3,B4)-COUNTIFS($B$3:$E$3,B4,$B4:$E4,B4),COUNTIF($B4:$E4,B4)-
COUNTIFS($B$3:$E$3,B4,$B4:$E4,B4))

Whites Functions (W4:Z14)

Instructions: Copy the functions below into their respective cells in the whites column. Then copy 
the first whites row to the rest of the whites rows (down to the bottom of your game-board.)

Copy the algorithm below into W4.

=$O4

Copy the algorithm below into X4.

=IF($P4=1,IF($C4=$B4,IF(AND($S4=2,$T4=2),1,IF($K4=1,1,0)),1),$P4)

Copy the algorithm below into Y4.

=IF($Q4=1,IF($D4=$B4,IF(AND($S4=2,$U4=2),1,IF($K4=1,1,0)),IF($D4=$C4,IF(AND($T4=2,$U4
=2),1,IF($L4=1,1,0)),1)),$Q4)

Copy the algorithm below into Z4.

=IF($R4=1,IF($E4=$B4,IF(AND($S4=2,$V4=2),1,IF($K4=1,1,0)),IF($E4=$C4,IF(AND($T4=2,$V4=
2),1,IF($L4=1,1,0)),IF($E4=$D4,IF(AND($U4=2,$V4=2),1,IF($M4=1,1,0)),1))),$R4)

Black & Whites Cell Functions (F4:I14)

Instructions: Copy the functions below into their respective cells in the black & whites columns 
(F4:I14). Then copy the first black & whites row to the rest of the rows in the black & whites columns 
(down to the bottom of your game-board.)

Copy the algorithm below into F4.

=IF(SUM(K4:N4)>0,1,IF(SUM(W4:Z4)>0,0,3))

Copy the algorithm below into G4.

=IF(F4=3,3,IF(F4=1,IF(SUM(K4:N4)>1,1,IF(SUM(W4:Z4)>0,0,3)), IF(SUM(W4:Z4)>1,0,3)))

Copy the algorithm below into H4.

=IF(G4=3,3,IF(G4=1,IF(SUM(K4:N4)>2,1,IF(SUM(W4:Z4)>0,0,3)),IF(F4=1,IF(SUM(W4:Z4)>1,0,3)
,IF(SUM(W4:Z4)>2,0,3))))

Copy the algorithm below into I4.

=IF(H4=3,3,IF(H4=1,IF(SUM(K4:N4)>3,1,IF(SUM(W4:Z4)>0,0,3)),IF(G4=1,IF(SUM(W4:Z4)>1,0,3
),IF(F4=1,IF(SUM(W4:Z4)>2,0,3),IF(SUM(W4:Z4)>3,0,3)))))

Conditional Rules

Instructions: Copy the Conditional Formatting functions below into the rules of the Conditional 
Format dialog box (ruler:Home, menu pick: Managing Rules..., select rule set: This Worksheet.)

Conditional Formatting Rule Format Applies to:
1 =or(isblank($B4), isblank($C4), isblank($D4),isblank($E4)) tan on tan =$F$4:$I$14
2 cell value = 3 tan on tan =$F$4:$I$14
3 cell value = 1 black on dk-grey =$F$4:$I$14
4 cell value = 0 lt grey on white =$F$4:$I$14
5 =and(isblank($K$3),not(countif($I4:$I14,1))) blue on blue =$B$3:$E$3
6 =and($F4=1,$G4=1,$H4=1,$I4=1) yellow =$B$4:$E$14
7 =isblank($K$3) white on white =$K$1:$AA$14
8 =isblank($K$3) tan on tan =$B$2:$H$2

Pre-whites Functions

The Pre-Whites functions in cells (O4:R14) are explained below in the detailed comments in blue and 
red. The colors are simply meant to separate the formatting of the logical if-statements.

Column O - 

IF(B$3=B4, 0, ;if B is black, then make it not-white...
IF(AND(B4=$C$3, NOT($C$3=$C4)), ;else, if the number is used in C, and C is not black,...

1, ;then make it white.
IF(AND($B4=$D$3, NOT($D$3=$D4)), ;else, if the # is used in D, and D is not black,...

1, ;then make it white.
IF(AND($B4=$E$3, NOT($E$3=$E4)), ;else, if the # is used in E, and E is not black,

1, ;then make it white.
0)))) ;else, make it not white.

Column P - 

=IF(C$3=C4, 0, ;if C is black, then make it not-white...
IF(AND(C4=$B$3, NOT($B$3=$B4)), ;else, if the number is used in B, and B is not black ...

1, ;then make it white.
IF(AND(C4=$D$3, NOT($D$3=$D4)), ;else, if the number is used in D, and D is not black,

1, ;then make it white.
IF(AND(C4=$E$3, NOT($E$3=$E4)), ;else, if the # is used in E, and E is not black, 

1,  ;then make it white.
0)))) ;else, make it not white.

Column Q - 

=IF(D$3=D4, 0, ;if D is black, then make it not-white...
IF(AND(D4=$B$3, NOT($B$3=$B4)), ;else, if the number is used in B, and B is not black,...

1, ;then make it white.
IF(AND(D4=$C$3, NOT($C$3=$C4)), ;else, if the number is used in C, and C is not black, 

1, ;then make it white.
IF(AND(D4=$E$3, NOT($E$3=$E4)), ;if the number is used in E, and E is not black,

1,  ;then make it white.
0)))) ;else, make it not white.

Column R - 

=IF(E$3=E4, 0, ;if E is black, then make it not-white...
IF(AND(E4=$B$3, NOT($E$3=$E4), NOT($B$3=$B4)), ;else, the number is used in B,...

1, ;and if  and B is not black, then make it white.
IF(AND(E4=$C$3, NOT($E$3=$E4), NOT($C$3=$C4)), ;else, if the number is used in C,...

1, ;and C is not black, then make it white.
IF(AND(E4=$D$3, NOT($E$3=$E4), NOT($D$3=$D4)),  ;else, if the number is used in D,...

1,  ;and D is not black, then make it white.
0)))) ;else, make it not white.

Multiple Functions

The Multiple functions in cells (S4:V14) are explained below in the detailed comments in blue  
and red. The colors are simply meant to separate the formatting of the statements.

=MIN( ;take the lesser of the two below:
COUNTIF($B$3:$E$3,B4) - ;the count of the number in Main Row, minus...
COUNTIFS($B$3:$E$3,B4,$B4:$E4,B4),     ;the count of the number that are blacks,
COUNTIF($B4:$E4,B4) - ;the count of the number in row four, minus...
COUNTIFS($B$3:$E$3,B4,$B4:$E4,B4)     ;the count of the number that are blacks
)

This algorithm determines how many extra occurrences of a given number there are in two 
compared rows besides matching blacks. 

It does this by first determining how many extra occurrences of a number there are in the 
master row minus the matching blacks, and then determining how many of that number there 
are in the second row besides matching blacks. By taking the minimum of these two results,  
one gets the quantity of extra occurrences of the number that are in both rows. 

This algorithm uses two Excel functions COUNTIF and COUNTIFS:

COUNTIF – receives a range of cells and compares it with a value. Then it returns the number  
of times it finds that value among the cells it checked. In this application, we are counting the  
number of times the number that was guessed occurs in a row of four cells.

COUNTIFS – this is very much like COUNTIF, except that COUNTIFS makes the comparison 
of multiple COUNTIF's and returns the number of times all of the comparisons are true. (In 
other words, COUNTIFS does an AND of multiple COUNTIF's.)

Whites Functions
(The White functions are in cells W4:Z14.)

Column W -      

=$O4 ;for the first cell, just use the result of the pre-white calculation. (mults don't matter)

Column X - 

=IF($P4=1, ;is the number in C is white?
IF($C4=$B4, ;yes, but is it the same number as in the B cell? (a multiple?)

IF(AND($S4=2,$T4=2), ;yes, but is both D and E of the Main row the same #?
1, ;yes, then C is white.
IF($K4=1,1,0)), ;no, then C is white only if B is a black.

1), ;no, then C is white.
$P4) ;no, then leave C not-white.

Column Y - 

=IF($Q4=1, ;is the number in D a white?
IF($D4=$B4, ;yes, but is it the same number as in the B cell? (a multiple?)

IF(AND($S4=2,$U4=2), ;yes, but is both C and E of the Main row the same #?
1, ;yes, then D is white.
IF($K4=1,1,0)), ;no, then D is white only if B is a black.

IF($D4=$C4, ;no, but is it the same number as in the C cell?
IF(AND($T4=2,$U4=2), ;yes, but is B and E of the Main row the same #?

1, ;yes, then D is white.
IF($L4=1,1,0)), ;no, then D is white only if C is a black

1)), ;no, then D is white.
$Q4) ;no, then leave D not-white.

Column Z - 

=IF($R4=1, ;is the number in E a white?
IF($E4=$B4, ;yes, but is it the same number as in the B cell? (a multiple?)

IF(AND($S4=2,$V4=2), ;yes, but is both C and E of the Main row the same #?
1, ;yes, then E is white.
IF($K4=1,1,0)), ;no, then E is white only if B is a black.

IF($E4=$C4, ;no, but is it the same number as in the C cell?
IF(AND($T4=2,$V4=2), ;yes, but is B and D of the Main row the same #?

1, ;yes, then E is white.
IF($L4=1,1,0)), ;no, then E is white only if C is a black

IF($E4=$D4, ;no, but is it the same number in the D cell?
IF(AND($U4=2,$V4=2), ;yes, but is B and C of the Main row the same #?

1, ;yes, then E is white
IF($M4=1,1,0)), ;no, then E is white only if D is a black.

1))), ;no, then E is white.
$R4) ;no, then leave E not-white.

Black & White Cell Functions

The Black & Whites functions in cells (F4:I14) are explained below in the detailed comments in blue 
and red. The colors are simply meant to separate the formatting of the logical if-statements.

Column F - 

=IF (SUM(K4:N4)>0, ;are there any blacks?
1, ;yes, then mark it black.
IF (SUM(W4:Z4)>0, ;are there any whites?

0, ;yes, then mark it white.
3)) ;otherwise make it a 3 for formatting null.

Column G - 

=IF (F4=3, ;is the first column blank?
3, ;yes, then this column is also blank. 
IF (F4=1, ;no, then is there a black already?

IF(SUM(K4:N4)>1, ;yes, then are there any more blacks?
1, ;yes, then mark it black.
IF(SUM(W4:Z4)>0,0,3)), ;no, make it white if there are whites, otherwise make it a 3.

IF(SUM(W4:Z4)>1,0,3))) ;no, then make it white if there are more whites, else make it a 3.

Column H - 

=IF(G4=3, ;is the first column blank?
3, ;yes, then this column is also blank. 
IF(G4=1, ;no, then are there two blacks already?

IF(SUM(K4:N4)>2, ;yes, then are there any more blacks?
1, ;yes, then mark it black.
IF(SUM(W4:Z4)>0,0,3)), ;no, make it white if there are whites, otherwise make it a 3.

IF(F4=1, ;no, then is there one black and one white, or two whites?
IF(SUM(W4:Z4)>1,0,3), ;yes, then make it white if there are >1 whites, else make it a 3.
IF(SUM(W4:Z4)>2,0,3)))) ;no, then make it white if there are >2 whites, else make it a 3.

Column I - 

=IF(H4=3, ;is the first column blank?
3, ;yes, then this column is also blank. 
IF(H4=1, ;no, then are there three blacks already?

IF(SUM(K4:N4)>3, ;yes, then are there any more blacks?
1, ;yes, then mark it black.
IF(SUM(W4:Z4)>0,0,3)), ;no, make it white if there are whites, otherwise make it a 3.

IF(G4=1, ;no, then are there two blacks and one white?
IF(SUM(W4:Z4)>1,0,3), ;yes, then make it white if there are more whites, else make it 3.
IF(F4=1, ;no, then is there one black and two whites?

IF(SUM(W4:Z4)>2,0,3), ;yes, then make it a white if 3 whites, else make it 3.
IF(SUM(W4:Z4)>3,0,3))))) ;no, then make it a white if 4 whites, else make it 3.

